Dateifunktionen Teil 1
Versionen: Access 2007, 2003, 2002/XP, 2000 und 97

In Ihren Datenbanken benötigen Sie immer wieder Funktionen für die Arbeit mit externen Dateien, um beispielsweise vor einem Import festzustellen, ob eine Datei existiert, um Laufwerk, Dateiname oder Erweiterung aus einem Pfad zu extrahieren oder um Dateinamen in Exportmakros zu setzen. In diesem zweiteiligen Tipp der Woche stellen wir Ihnen einige nützliche Funktionen und Prozeduren für derartige Aufgaben vor, erklären die VBA-technische Umsetzung im Detail und zeigen deren Einsatz anhand kleiner Beispiele. Wenn Sie die vorgestellten Funktionen in Ihren Datenbanken nutzen möchten, kopieren Sie sie einfach aus dem Newsletter in ein VBA-Modul.

Um den Dateinamen der aktuellen Datenbank zu ermitteln, nutzen Sie die Funktion "GetDatabaseName()":

Function GetDatabaseName() As String
  Dim db As DATABASE, L As Integer, strPath As String

  Set db = CurrentDb()
  strPath = db.Name
  L = Len(strPath)
  While Mid$(strPath, L, 1) <> "\" And L > 0
    L = L - 1
  Wend
  If L > 1 Then
    strPath = Mid$(strPath, L + 1)
  Else
    strPath = ""
  End If
  GetDatabaseName = strPath

End Function

Die Funktion liefert als Ergebnis den Dateinamen der aktuell geöffneten Datenbank, also zum Beispiel "Adressen.mdb" oder "Bestellungen.accdb". Diesen Dateinamen können Sie zum Beispiel in Formularen anzeigen oder in Berichten im Seitenfuß ausgeben, um so beispielsweise bei telefonischen Rückfragen schnell ermitteln zu können, mit welcher Datenbank gearbeitet wird oder aus welcher Datenbank der Bericht stammt. Dazu legen Sie im Formular zum Beispiel im Formularkopf oder im Bericht ein ungebundenes Textfeld an und weisen diesem Feld in der Eigenschaft "Steuerelementeinhalt" die folgende Formel zu:

=GetDatabaseName()

Basis der Funktion ist die Eigenschaft "Name" des "Database"-Objektes. Das "Database"-Objekt repräsentiert die aktuell geöffnete Datenbank. Die Eigenschaft "Name" beinhaltet Laufwerk, Pfad und Dateiname der Datenbank, also zum Beispiel "D:\ Daten\ Access\ Adresen.mdb". "GetDatabasName()" extrahiert aus diesen Angaben den Dateinamen: Dazu liefert die Funktion "CurrentDb()"einen Verweis auf die aktuell geöffnete Datenbank, den wir der Objektvariablen "db" zuweisen. Die Variable "strPath" beinhaltet jetzt den kompletten Pfad aus der Eigenschaft "Name", also beispielsweise "D:\Daten\Access\Adressen.Mdb". In einer Schleife durchlaufen wir den Pfad von hinten nach vorne, bis der erste Backslash "\" gefunden wurde - alles rechts vom letzten Backslash ist dann der Dateiname. Die Variable "L" wird dazu zunächst mit der Anzahl Zeichen initialisiert, die der Pfad lang ist, und zeigt anfangs somit auf das letzte Zeichen. Die Funktion "Mid$(strPath, L, 1)" liefert das Zeichen, dessen Position "L" angibt. Ist das nicht der Backslash, subtrahieren wir "1" von der Variablen "L" und bewegen uns so von rechts nach links im Pfad. Zeigt "L" auf einen Backslash, ist die Bedingung für die While-Schleife erfüllt und es wird mit den Anweisungen nach "Wend" weitergemacht.

Damit es keine Laufzeitfehlermeldung gibt, wenn der Pfad keinen Backslash enthält, setzen wir zusätzlich die Bedingung "L > 0" ein. Dadurch ist sichergestellt, dass "L" nicht bis auf "0" heruntergezählt und womöglich ein "Mid$(strPath,0,0)" ausgeführt wird. Diese Anweisung würde eine Laufzeitfehlermeldung verursachen, da es innerhalb von Zeichenketten keine Position "0" gibt. Anschließend prüfen wir zunächst, ob der Zeiger "L" größer "1" ist - wenn ja, wurde ein Backslash gefunden und wir können einen Dateinamen extrahieren. Dazu setzen wir die Funktion "Mid$()" ab der Position "L+1" ein, also ab dem Zeichen direkt nach dem letzten Backslash. Abschließend weisen wir den Inhalt der Variablen "strPath" und somit den Dateinamen der Datenbank als Funktionsergebnis zu - fertig!

Die Funktion "GetDatabaseName()" können Sie natürlich nicht nur in Formularen und Berichten, sondern auch in VBA einsetzen. Wenn Sie zum Beispiel den Dateinamen nicht kontinuierlich in einem Formular anzeigen sondern durch einen Klick auf eine Schaltfläche melden lassen möchten, würde eine Ereignisprozedur "Beim Klicken" für eine entsprechende Schaltfläche wie folgt aussehen:

Sub btnGetDBName_Click ()

  MsgBox GetDatabaseName()

End Sub

Um nur den Pfad der aktuellen Datenbank zu ermitteln, nutzen Sie die Funktion "GetDatabasePath()":

Function GetDatabasePath() As String
  Dim db As DATABASE, L As Integer, strPath As String

  Set db = CurrentDb()
  strPath = db.Name
  L = Len(strPath)
  While Mid$(strPath, L, 1) <> "\" And L > 0
    L = L - 1
  Wend
  If L > 1 Then
    strPath = Left$(strPath, L)
  Else
    strPath = ""
  End If
  GetDatabasePath = strPath

End Function

Die Funktion arbeitet ähnlich wie "GetDatabaseName()". Sie basiert ebenfalls auf der "Name"-Eigenschaft des "Database"-Objektes, liefert aber als Ergebnis den Pfad ohne Dateinamen. Wenn Sie zum Beispiel Textdateien per VBA aus dem Installationsverzeichnis der Datenbank importieren möchten, ermitteln Sie den Dateinamen für die Textdatei wie folgt:

strImportDatei = GetDatabasePath() & "Import.txt"

"GetDabasePath()" ist weiterhin zum Beispiel recht nützlich in Makros einzusetzen, die über "TransferText" Daten aus externen Datendateien importieren. Hier müssen Sie als Parameter immer einen Dateinamen inklusive Laufwerk und Pfad angeben. Wird die Datenbank einmal in einem anderen als dem hier ursprünglich spezifizierten Verzeichnis abgelegt, kann das Makro ohne Änderung nicht mehr ausgeführt werden, es kommt zu einer Fehlermeldung. Setzen Sie hingegen als Dateinamen im Makro die Formel "=GetDatabasePath() & "Import.txt" ein, spielt es keine Rolle mehr, in welchem Verzeichnis sich die Datenbank befindet - die Importdateien werden immer zuverlässig im Datenbankverzeichnis gefunden.

Im Großen und Ganzen ist die Funktion "GetDatabasePath()" aufgebaut wie "GetDatabaseName()". Der Unterschied liegt lediglich darin, dass hier anstelle des Dateinamens rechts vom letzten Backslash nun den Pfad links inklusive des Backslash als Funktionsergebnis geliefert wird.

Außer mit Pfad und Dateiname der aktuellen Datenbank haben Sie regelmäßig mit externen Dateien für den Import/Export zu tun, bei denen ebenfalls Einzelinformationen aus einem kompletten Pfad benötigt werden. Wenn Sie beispielsweise vor dem Überschreiben einer Exportdatei eine Sicherheitskopie anlegen möchten, benötigen Sie zunächst den Dateinamen und müssen ihm eine abweichende Erweiterung wie beispielsweise ".BAK" zuweisen. Oder Sie möchten sicherstellen, dass die Datenbank immer auf einem bestimmten Laufwerk mit dem Label "Data" installiert ist. Dazu muss der Laufwerksbuchstabe extrahiert werden. Für diese und ähnliche Aufgaben nutzen Sie die folgenden Funktionen:

Function GetDrive (strPath As String) As String

  If InStr(strPath, ":") <> 0 Then
    GetDrive = Left$(strPath, InStr(strPath, ":"))
  ElseIf InStr(strPath, "\\") <> 0 Then
    GetDrive = Left$(strPath, InStr(strPath, "\\") + 1)
  Else
    GetDrive = ""
  End If

End Function

Die Funktion "GetDrive()"extrahiert aus einem kompletten Pfad den Laufwerksbuchstaben wie beispielsweise "C:" oder "D:" oder liefert bei Netzwerkpfaden den angegebenen Server als Ergebnis. Dazu prüft die Funktion zunächst, ob der angegebene Pfad einen Doppelpunkt oder den bei Netzwerkpfaden zur Trennung verwendeten Doppelslash ("Servername\\Volume\Verzeichnis" oder ähnlich) beinhaltet. Dazu setzen wir die Access-interne Funktion "InStr()" ein. Diese durchsucht die als erstem Parameter angegebene Zeichenkette nach dem als zweitem Parameter angegebenen Zeichen. Das Ergebnis ist "0", wenn das Zeichen nicht enthalten ist oder andernfalls die Position, an der das Zeichen gefunden wurde. Ist das Ergebnis bei "InStr(strPath, ":")" ungleich "0", ist ein Doppelpunkt und somit auch ein normaler Laufwerksbuchstabe vorhanden. "InStr()" kann aber auch nach mehr als einem Zeichen suchen. Das Ergebnis ist dann die Position des 1. Zeichens der Übereinstimmung. Das machen wir uns in der zweiten Abfrage " ElseIf InStr(strPath, "\\") <> 0 Then..." zunutze. Ist das Ergebnis hier ungleich "0", handelt es sich um einen Netzwerkpfad und wir liefern den Namen des Servers inklusive des Doppelslash als Ergebnis zurück. Waren beide Prüfungen nicht erfolgreich, setzen wir das Ergebnis auf "", also eine leere Zeichenkette.

In Ihren Datenbanken benötigen Sie immer wieder Funktionen für die Arbeit mit externen Dateien, um beispielsweise vor einem Import festzustellen, ob eine Datei existiert, um Laufwerk, Dateiname oder Erweiterung aus einem Pfad zu extrahieren oder um Dateinamen in Exportmakros zu setzen. In diesem zweiteiligen Tipp der Woche stellen wir Ihnen einige nützliche Funktionen und Prozeduren für diese Aufgaben vor. Wenn Sie die vorgestellten Funktionen in Ihren Datenbanken nutzen möchten, kopieren Sie sie einfach aus dem Newsletter in ein VBA-Modul.

Um den Namen eines Laufwerks zu ermitteln, kann die auf "GetDrive()" basierende Funktion "GetLabel()" eingesetzt werden, die den Namen eines Laufwerkes (das sogenannte "Volume Label") als Ergebnis liefert:

Function GetLabel(strPath As String) As String
    Dim strDrive As String

    strDrive = GetDrive(strPath)
    GetLabel = Dir$(strDrive, vbVolume)

End Function

Damit können Sieprüfen und/oder sicherstellen, dass eine Datenbank immer auf einem bestimmten Laufwerk installiert ist. Eine Abfrage auf korrekte Installation könnte so aussehen:

strLabel = GetLabel(GetDatabasePath())
If strLabel <> "Data" Then
  Beep
  MsgBox "Datenbank auf falschem Laufwerk installiert!"
  DoCmd.Quit
End If

Die Funktion ermittelt zunächst den Laufwerksbuchstaben aus dem als Parameter übergebenen Pfad und weist ihn der Variablen "strDrive" zu. Für dieses Laufwerk wird dann ein "Dir$()" ausgeführt. Diese Funktion liefert normalerweise Dateinamen aus einem Verzeichnis zurück. Über einen zweiten Parameter können Sie hier aber auch festlegen, welche Dateiattribute die Dateien aufweisen sollen. Verwenden Sie das Attribut "vbVolume", wird der Name (das Volume Label) des betreffenden Laufwerkes als Ergebnis geliefert.

Nun zu zwei Funktionen, um Dateinamen abzuschneiden und einen Pfad zu ermitteln. Die Funktion "GetPath()" ist ein Auszug aus der oben vorgestellten Funktion "GetDatabasePath()". Sie liefert als Ergebnis Laufwerk und Pfad inklusive letztem Backslash für den als Parameter übergebenem kompletten Pfad, schneidet also den Dateinamen ab:

Function GetPath (strPath As String) As String
  Dim L As Integer

  GetPath = ""
  If InStr(strPath, "\") = 0 Then Exit Function

  L = Len(strPath)
  While Mid$(strPath, L, 1) <> "\" And L > 0
    L = L - 1
  Wend
  If L > 1 Then GetPath = Left$(strPath, L)

End Function

Diese Funktion können Sie beispielsweise in Kombination mit der gleich vorgestellten Funktion "GetName()" einsetzen, um einen Dateinamen für eine Sicherheitskopie zusammenzusetzen. Das folgende Beispiel verwendet in "strFilename" einen kompletten Pfad "D:\Daten\Access\Adressen.mdb":

'D:\Daten\Access\
strPath = GetPath(strFileName)
'Adressen
strFilename = GetName(strFilename)
'D:\Daten\Access\Adressen.bak
strBAKFile = strPath + strFilename + ".bak"
FileCopy strFilename, strBAKFile

Analog zu "GetPath()" arbeitet die Funktion "GetFilename()". Sie liefert als Ergebnis nur den Dateinamen inklusive Erweiterung aus dem als Parameter übergebenen Pfad:

Function GetFilename (strPath) As String
  Dim L As Integer

  If InStr(strPath, "\") <> 0 Then
    L = Len(strPath)
    While Mid$(strPath, L, 1) <> "\" And L > 0
      L = L - 1
    Wend
    If L > 1 Then
      strPath = Mid$(strPath, L + 1)
    Else
      strPath = ""
    End If
  Else
    strPath = ""
  End If

  GetFilename = strPath

End Function

In der Praxis ist diese Funktion hilfreich, um zum Beispiel Pfade für das Kopieren von Dateien zusammenzusetzen. Das folgende Beispiel verwendet in "strFilename" einen kompletten Pfad "D:\Daten\Access\import.txt":

strQuelle = strFilename
strZiel = "E:\Daten\Datsi\" + GetFilename(strQuelle)
FileCopy strQelle, strZiel

Hier kopieren wir eine Importdatei auf ein anderes Laufwerk in ein spezielles Verzeichnis für die Datensicherung. Der Pfad "E:\Daten\Datsi\" ist konstant vorgegeben, den Dateinamen (hier "Import.txt") ermitteln wir per "GetFilename()" aus dem Quellpfad und setzen dann beides als Zielpfad zusammen.

Für einige Aktionen wie beispielsweise dem Anlegen von Sicherheitskopien benötigen Sie nur den Dateinamen ohne Erweiterung, also zum Beispiel "Import" aus "Import.txt". Hierfür können Sie die Funktion "GetName()" nutzen:

Function GetName (strPath As String) As String
  Dim L As Integer

  If InStr(strPath, "\") <> 0 Then
    L = Len(strPath)
    While Mid$(strPath, L, 1) <> "\" And L > 0
      L = L - 1
    Wend
    If L > 1 Then strPath = Mid$(strPath, L + 1)
  End If

  If InStr(strPath, ".") <> 0 Then
    L = Len(strPath)
    While Mid$(strPath, L, 1) <> "." And L > 0
      L = L - 1
    Wend
    If L > 1 Then
      strPath = Left$(strPath, L - 1)
    Else
      strPath = ""
    End If
  End If

  GetName = strPath

End Function

Im ersten Teil wird hier zunächst der komplette Dateiname extrahiert. Diese Routine entspricht somit der Funktion "GetFilename()", die wir weiter oben vorgestellt haben. Der zweite Teil der Funktion sucht nun rückwärts im Dateinamen nach der Position des Punktes, der Name und Erweiterung trennt. Alles links vom Punkt wird dann als Funktionsergebnis zurückgegeben.

Ein Beispiel für den Einsatz dieser Funktion haben wir oben bei der Funktion "GetPath()" bereits aufgeführt. Dort wurde "GetName()" für das Zusammensetzen von Pfaden für die Datensicherung mit abweichender Erweiterung eingesetzt.

